

A formal proof of Pick’s Theorem

JOHN HARRISON

Intel Corporation, JF1-13, 2111 NE 25th Avenue, Hillsboro OR 97124, U.S.A.
Email: johnh@ichips.intel.com

Received 20 July 2010; revised 15 December 2010


Pick’s Theorem relates the area of a simple polygon with vertices at integer lattice points to the number of lattice points in its inside and boundary. We describe a formal proof of this theorem using the HOL Light theorem prover. As sometimes happens for highly geometrical proofs, the formalisation turned out to be more work than initially expected. The difficulties arose mostly from formalising the triangulation process for an arbitrary polygon.

1. Introduction

We start with some definitions, whose formal counterparts will be given later. Throughout this paper we call a point $(x, y) \in \mathbb{R}^2$ of the plane an *integer lattice point*, or simply a *lattice point*, if and only if both x and y are integers. We define a *lattice polygon* to be a polygon all of whose vertices are at integer lattice points. Pick’s Theorem (Pick 1899) states that given a simple lattice polygon, if I is the number of integer lattice points in its inside and B the number of integer lattice points on its boundary, then the area of the polygon is given by

$$A = I + B/2 - 1$$

For example, the following diagram shows a lattice polygon, with the lattice points in its vicinity marked with black spots. There are 30 lattice points in the inside of the polygon and 22 on its boundary, so according to Pick’s theorem the area of the polygon is $30 + 22/2 - 1 = 40$.



To state Pick’s Theorem more precisely, we need to be a bit more careful about what we consider a polygon to be and what constitutes its ‘inside’ and ‘boundary’. We will pay more attention to these details later. For now, we just recall that a *curve* or *path* in the plane can be considered formally simply as a continuous function $\gamma : [0, 1] \rightarrow \mathbb{R}^2$ out of the unit interval, and a *polygonal path* is the special case of a piecewise linear continuous function. A path is said to be *closed* if its starting and finishing points are the same

($\gamma(0) = \gamma(1)$), and *simple* if it does not otherwise self-intersect (that is, $\gamma(s) = \gamma(t)$ implies either $s = t$ or $\{s, t\} = \{0, 1\}$). We can identify a *simple polygon* with the image $\gamma[0, 1]$ of the unit interval under a simple closed polygonal path γ .

This paper describes a formalisation of Pick’s Theorem using the HOL Light theorem prover (Harrison 1996). The proof turned out to be somewhat harder work than expected. Using the crude metric of ‘lines of proof script’, it seems to be almost as challenging as the Prime Number Theorem and considerably harder than Dirichlet’s Theorem on primes in arithmetic progression.

Theorem	Lines of proof
Dirichlet’s Theorem (Harrison 2009a)	2082 lines
Pick’s Theorem (the present paper)	3709 lines
Prime Number Theorem (Harrison 2009b)	4314 lines

Charming and surprising as Pick’s Theorem is, one would normally consider it a very straightforward result in comparison with those two jewels of 19th century mathematics. What accounts for the difficulty of formalising Pick? Of course, it could be just a reflection of our own ineptitude. However, we prefer to believe that there is an intrinsic difficulty in formalising many geometric proofs in stating and proving in a formal way various properties that seem intuitively obvious ‘by eye’. In fact, the main difficulties with formalising Pick’s Theorem are not at all specific to that theorem, but are connected with the process of ‘triangulating’ a general polygon, as is done in the informal proof we were using as a model.

2. Preliminary definitions and lemmas

We will begin by describing the formalisation of the various concepts involved in stating Pick’s Theorem. The material connected with paths is fairly standard, but the precise definition of ‘inside’ is perhaps more interesting.

Lattice points

We define the lattice points as ‘integral vectors’, that is, vectors all of whose components are integers. The definition is made for the more general Euclidean space \mathbb{R}^N using the technical setup described in Harrison (2005), but we will only use the special case \mathbb{R}^2 in what follows.

```
|- integral_vector x <=>
    <math>\forall i. 1 \leq i \wedge i \leq \text{dimindex}(:N) \Rightarrow \text{integer}(x\ \$i)</math>
```

Paths and polygons

As noted, we consider a path to be just a continuous function out of the unit interval. Actually, the HOL Light formalisation uses the unit interval in the type \mathbb{R}^1 , which is

technically different, though isomorphic to, the unit interval on \mathbb{R} . The elements `vec 0` and `vec 1` are the endpoints of this interval.

```
|- path g ↔ g continuous_on interval[vec 0,vec 1]
```

For most purposes, we want to forget the details of the parametrisation using the arbitrary interval $[0, 1]$, so we define various natural abbreviations as follows:

```
|- pathstart g = g(vec 0)
|- pathfinish g = g(vec 1)
|- closed_path g ↔ pathstart g = pathfinish g
|- path_image g = IMAGE g (interval[vec 0,vec 1])
```

One important special case of a path is the straight-line path from point a to point b . The functions `drop` : $\mathbb{R}^1 \rightarrow \mathbb{R}$ and `lift` : $\mathbb{R} \rightarrow \mathbb{R}^1$ are the inverse bijections between the type \mathbb{R}^1 and \mathbb{R} .

```
|- linepath(a,b) = λx. (&1 - drop x) % a + drop x % b
```

It is often convenient to stick together two paths to make a new path. To retain the canonical parametrisation, we allocate the two intervals $[0, 1/2]$ and $[1/2, 1]$ to scaled versions of the the two components. Formally, we have:

```
|- g1 ++ g2 = (λx. if drop x <= &1 / &2
                  then g1(&2 % x)
                  else g2(&2 % x - vec 1))
```

We define an *arc* to be a path that does not self-intersect anywhere:

```
|- arc g ↔
  path g ∧
  ∀x y. x IN interval[vec 0,vec 1] ∧
        y IN interval[vec 0,vec 1] ∧
        g x = g y
        ⇒ x = y
```

And a *simple path* is defined to be one that may intersect only at its endpoints:

```
|- simple_path g ↔
  path g ∧
  ∀x y. x IN interval[vec 0,vec 1] ∧
        y IN interval[vec 0,vec 1] ∧
        g x = g y
        ⇒ x = y ∨ x = vec 0 ∧ y = vec 1 ∨ x = vec 1 ∧ y = vec 0
```

Note that every arc is therefore also a simple path, and various other straightforward relationships hold:

```
|- arc g ⇒ simple_path g
|- simple_path g ⇒ path g
|- arc g ↔ simple_path g ∧ ¬(pathfinish g = pathstart g)
```

We consider a polygonal path to be one defined by a list of vertices $[v_0; v_1; \dots; v_n]$. Intuitively, this is the path that proceeds through those vertices in order via straight-line segments. It is defined by list recursion as a succession of `linepaths`:

```

|- polygonal_path [] = linepath(vec 0,vec 0) ^
  polygonal_path [a] = linepath(a,a) ^
  polygonal_path [a;b] = linepath(a,b) ^
  polygonal_path (CONS a (CONS b (CONS c 1))) =
    linepath(a,b) ++ polygonal_path(CONS b (CONS c 1))

```

We will seldom be concerned with the cases of empty or singleton lists, but they are defined for the sake of regularity. In the case of the empty list, the choice of the corresponding `linepath` as the constant function whose image is $\{0\}$ is made only to avoid technical restrictions on the theorem about the image of a polygonal path under a linear mapping, and has no deep significance.

Inside and outside

Pick's Theorem talks about the 'inside' or 'interior' of a polygon. First, note that there is already an established topological theory in HOL Light, which includes the interior, closure and frontier (boundary) of a set:

```

|- interior s = {x | ∃t. open t ^ x IN t ^ t SUBSET s}
|- closure s = s UNION {x | x limit_point_of s}
|- frontier s = (closure s) DIFF (interior s)

```

Despite the disparate definitions, the interior and closure obey well-known dualities:

```

|- closure s = (:real^N) DIFF (interior ((:real^N) DIFF s))
|- closure((:real^N) DIFF s) = (:real^N) DIFF interior(s)

```

However, when one talks about the 'inside' or 'interior' of a polygon in Pick's Theorem, one means not the interior of the polygonal path itself (that would be the empty set since it has zero thickness) but rather of the region it encloses. So the fundamental problem is to define that concept. A clearly relevant result is the Jordan Curve Theorem, which asserts that a simple closed curve divides the plane into an 'inside' and 'outside'. At least, that is how one thinks of it intuitively, but the traditional formal statements simply observe that the complement of such a curve has two connected components, each of which has the curve as its frontier, and exactly one of which is bounded. For a discussion of the formal proof of this statement, see Hales (2007b).

```

|-  $\forall c:\text{real}^1 \rightarrow \text{real}^2.$ 
  simple_path c  $\wedge$  pathfinish c = pathstart c
   $\Rightarrow \exists \text{ins out}.$ 
     $\neg(\text{ins} = \{\}) \wedge \text{open ins} \wedge \text{connected ins} \wedge$ 
     $\neg(\text{out} = \{\}) \wedge \text{open out} \wedge \text{connected out} \wedge$ 
    bounded ins  $\wedge$   $\neg$ bounded out  $\wedge$ 
    ins INTER out =  $\{\}$   $\wedge$ 
    ins UNION out =  $(:\text{real}^2)$  DIFF path_image c  $\wedge$ 
    frontier ins = path_image c  $\wedge$ 
    frontier out = path_image c

```

Intuitively, the components ‘ins’ and ‘out’ whose existence is asserted are thought of as the ‘inside’ and ‘outside’ of the curve. We might elect to introduce these notions simply by Skolemising these existential quantifiers. However, we prefer to define the concepts in a more general way that might be applicable in other situations. In particular, these apply to arbitrary sets in \mathbb{R}^N , not just paths in \mathbb{R}^2 (even if we might not be able to deduce analogously strong properties). We simply say that a point x is inside (respectively, outside) a set S if it is not in S and the connected component of $\mathbb{R}^N - S$ containing x is bounded (respectively, unbounded).

```

|- inside s = {x |  $\neg(x \text{ IN } s) \wedge$ 
  bounded(connected_component  $((:\text{real}^N)$  DIFF s) x)}
|- outside s = {x |  $\neg(x \text{ IN } s) \wedge$ 
   $\neg$ bounded(connected_component  $((:\text{real}^N)$  DIFF s) x)}

```

It is fairly straightforward to prove a collection of natural and straightforward properties of these concepts, for example:

```

|-  $\forall s.$  inside s INTER s =  $\{\}$ 
|-  $\forall s.$  inside s UNION outside s =  $(:\text{real}^N)$  DIFF s
|-  $\forall s.$  inside s =  $(:\text{real}^N)$  DIFF (s UNION outside s)
|-  $\forall s t.$  s SUBSET t  $\Rightarrow$  outside t SUBSET outside s
|-  $\forall s.$  bounded s  $\Rightarrow$  bounded $((:\text{real}^N)$  DIFF outside s)
|-  $\forall c_1 c_2.$  c INTER outside(c1 UNION c2) =  $\{\}$ 
   $\Rightarrow$  outside(c1 UNION c2) SUBSET outside(c1 UNION c)
|-  $\forall s.$  convex s  $\Rightarrow$  outside s =  $(:\text{real}^N)$  DIFF s
|-  $\forall s.$  bounded s  $\wedge$  convex s  $\Rightarrow$  inside(frontier s) = interior s


```

Although in general it might be rather challenging to prove some other significant properties in full generality, we can at least fairly straightforwardly use it to ‘Skolemise’ the Jordan Curve Theorem as discussed above.

```

|- !c:real^1->real^2.
  simple_path c ^ pathfinish c = pathstart c
  => ~(inside(path_image c) = {}) ^
      open(inside(path_image c)) ^
      connected(inside(path_image c)) ^
      ~(outside(path_image c) = {}) ^
      open(outside(path_image c)) ^
      connected(outside(path_image c)) ^
      bounded(inside(path_image c)) ^
      ~bounded(outside(path_image c)) ^
      inside(path_image c) INTER outside(path_image c) = {} ^
      inside(path_image c) UNION outside(path_image c) =
      (:real^2) DIFF path_image c ^
      frontier(inside(path_image c)) = path_image c ^
      frontier(outside(path_image c)) = path_image c
  
```

As will be explained later, one of the key ideas in the Pick proof is to divide the inside of a polygon into two pieces by a straight cut across its inside. Once again, we made an attempt to generalise this property from polygonal arcs to arbitrary ones. Suppose an arc c cuts across the inside of a simple closed curve, meeting it at points a and b , and we separate the original closed curve into two arcs c_1 and c_2 with the endpoints a and b . Our objective is to prove that the inside of the original curve is essentially cut into two pieces, one the inside of the curve defined by c_1 and c , the other the inside of the curve defined by c_2 and c .



We had originally hoped that given the Jordan Curve Theorem, all such natural extensions and variants (we sometimes call this one the ‘Jordan Triple Curve Theorem’) would be fairly trivial corollaries. But, in fact, this is the sort of theorem that is not considered very often in the literature, and we had some trouble finding a suitable proof to formalise. We eventually found a relatively straightforward 14-line proof in Whyburn (1964, 1.4, page 31). However, it was still quite hard work to formalise: in particular, the process of cutting paths into sub-arcs and glueing them together in different ways proved awkward. This is the final result as stated and proved in HOL Light; note that we merely assume that c has non-empty intersection with the inside of the curve defined by c_1 and c_2 (though this generalisation is trivial).

```

|- ∀c1 c2 c a b:real^2.
  ¬(a = b) ∧
  simple_path c1 ∧ pathstart c1 = a ∧ pathfinish c1 = b ∧
  simple_path c2 ∧ pathstart c2 = a ∧ pathfinish c2 = b ∧
  simple_path c ∧ pathstart c = a ∧ pathfinish c = b ∧
  path_image c1 INTER path_image c2 = {a,b} ∧
  path_image c1 INTER path_image c = {a,b} ∧
  path_image c2 INTER path_image c = {a,b} ∧
  ¬(path_image c INTER inside(path_image c1 UNION path_image c2) = {})
⇒ inside(path_image c1 UNION path_image c) INTER
  inside(path_image c2 UNION path_image c) = {} ∧
  inside(path_image c1 UNION path_image c) UNION
  inside(path_image c2 UNION path_image c) UNION
  (path_image c DIFF {a,b}) =
  inside(path_image c1 UNION path_image c2)
    
```


The eventual formal proof of this result was fully 788 lines long. We do not consider this as part of the proof of Pick's Theorem specifically, and have installed it with the general multivariate theories. For this reason, it was not included in the line-count of 3709 for Pick's Theorem. However, this categorisation is arguable, and including these further 788 lines in the count makes the Pick proof even longer than that of the Prime Number Theorem!

3. The proof of Pick's Theorem

Having established the necessary background to state and prove Pick's Theorem, we can now get to work. Our proof of the theorem is fairly standard, though it does not specifically follow any single informal source. It proceeds by establishing the theorem for increasingly general classes of polygon: first, a so-called 'elementary triangle', then an arbitrary lattice triangle and finally a general simple polygon. As will be seen, the first two steps are entirely routine and straightforward to formalise, while the last one presents significant difficulties.

Elementary triangles

An *elementary triangle* is one with vertices at lattice points but containing no other lattice points, either inside or on its boundary. The following picture shows some examples of elementary triangles.



Pick's Theorem for such a triangle simply asserts that each such triangle has area $1/2$. There are various relatively straightforward proofs of this result. The one we have formalised is based on linear transformations of the integer lattice.

Given two vectors A and B , we can consider them as defining the linear transformation of the plane $f : (x, y) \mapsto Ax + By$, where juxtaposition indicates scalar-vector multiplication. It is not hard to show that if the image under f of the set of integer lattice points is exactly this same set of integer lattice points, then the determinant of the matrix of f is ± 1 :

```
|- ∀f:real^N->real^N.
    linear f ∧ IMAGE f integral_vector = integral_vector
    ⇒ abs(det(matrix f)) = &1
```

Given an elementary triangle OAB , where we take O as the origin, one can show that the integer multiples of the two other vertices generate precisely the integer lattice points. The determinant in the previous theorem is precisely twice the area of the triangle formed by the three vertices, which therefore has area $1/2$.

In formalising this result, we simply use the convex hull of the set of vertices $\{a, b, c\}$ instead of the more elaborate concept of being ‘inside’ the triangle abc (though we can connect the two definitions easily enough later). Our characterisation of an elementary triangle is that the set of integer lattice points inside this convex hull is exactly the vertex set $\{a, b, c\}$. This actually takes in the possibility that the triangle is completely degenerate, i.e. that the three vertices are collinear. So the formal theorem has to take this into account.


```
|- ∀a b c:real^2.
    {x | x IN convex hull {a,b,c} ∧ integral_vector x} = {a,b,c}
    ⇒ measure(convex hull {a,b,c}) =
        if collinear {a,b,c} then &0 else &1 / &2
```

Arbitrary triangles

Next, we proceed inductively to establish the result for an arbitrary triangle with all its vertices at integer lattice points. If a triangle ABC is not elementary, then it must have a lattice point D either:

- On one of its sides, say AB , in which case we can subdivide it into triangle ADC and BCD .
- In its interior, in which case we can divide it into three triangles ADB , ADC and BDC .

The following diagram illustrates these possibilities for a few examples.



Although this is straightforward enough, we can make it even simpler, by reformulating things slightly so that the first case becomes a special case of the second, so we can avoid handling degenerate cases of the theorem itself separately. The key is the following general theorem about 'additivity' of a real-valued function defined on subsets of the plane.

```

|- ∀f:(real^2->bool)->real a b c d.
  (∀s t. compact s ∧ compact t
   ⇒ f(s UNION t) = f(s) + f(t) - f(s INTER t)) ∧
  ¬(a = b) ∧ ¬(a = c) ∧ ¬(b = c) ∧
  ¬affine_dependent {a,b,c} ∧ d IN convex_hull {a,b,c}
  ⇒ f(convex_hull {a,b,c}) =
    (f(convex_hull {a,b,d}) +
     f(convex_hull {a,c,d}) +
     f(convex_hull {b,c,d})) -
    (f(convex_hull {a,d}) +
     f(convex_hull {b,d}) +
     f(convex_hull {c,d})) +
    f(convex_hull {d})
  
```

In the inductive proof of Pick's Theorem for an arbitrary lattice triangle, we can apply this general result twice, once with f giving the number of lattice points in a set and once with it giving the measure (area) of a set. This results in an almost immediate proof of the following reformulation of Pick's theorem:

```

|- ∀a b c:real^2.
  integral_vector a ∧ integral_vector b ∧ integral_vector c
  ⇒ measure(convex_hull {a,b,c}) =
    &(CARD {x | x IN convex_hull {a,b,c} ∧ integral_vector x}) -
    (&(CARD {x | x IN convex_hull {b,c} ∧ integral_vector x}) +
     &(CARD {x | x IN convex_hull {a,c} ∧ integral_vector x}) +
     &(CARD {x | x IN convex_hull {a,b} ∧ integral_vector x})) / &2 +
    &1 / &2
  
```

It is straightforward to show that it is equivalent to the usual formulation with a proviso of non-degeneracy:

```

|- ∀a b c:real^2.
  integral_vector a ∧ integral_vector b ∧ integral_vector c
  ⇒ measure(convex_hull {a,b,c}) =
    if collinear {a,b,c} then &0
    else &(CARD {x | x IN interior(convex_hull {a,b,c}) ∧
      integral_vector x}) +
      &(CARD {x | x IN frontier(convex_hull {a,b,c}) ∧
      integral_vector x}) / &2 - &1
  
```

Arbitrary polygons

Again, we proceed inductively, showing that any polygon can be subdivided into two by a line joining two vertices and otherwise lying entirely in the inside. (This can also be used to drive an inductive proof that any polygon can be triangulated, and that is where we looked for a proof, though we do not explicitly deduce this general result.) The informal proof, essentially excerpted from Hales (2007a), seems relatively straightforward:

Pick the coordinate axis so that no two vertices have the same y coordinate. Let B be the lowest vertex on the polygon, and let A and C be adjacent to B . If AC is an interior diagonal, we draw the diagonal AC , forming a triangle ABC and a polygon without the vertex B . Otherwise, let D be a

vertex of the polygon at maximal distance from the line AC in the direction of B. Cut the polygon into two along the edge BD.

The first challenge is to formalise the ‘pick the coordinate axis’ step. In an earlier paper, Harrison (2009c), we described an extensive framework for such ‘without loss of generality’ reasoning, but to support the present proof, this had to be generalised from ‘first order’ concepts like points and lines to ‘higher order’ concepts like polygonal paths and lists of points. The following ‘WLOG lemma’ gives the schematic justification for picking a suitable transform f that enables us to assume that some particular object x has a nice property.

```


|- (∀x. ∃f y. transform f ∧ nice y ∧ f y = x)
  => ∀P. (∀f x. transform f => (P(f x) ⇔ P x)) ∧
        (∀x. nice x => P x)
  => ∀x. P x
    
```

In the applications considered in Harrison (2009c), the objects were always individual vectors, and the transformations were aimed at making one of these vectors have some particularly convenient property. In a typical example, one chooses a spatial translation to turn one point into the origin; in another one chooses an orthogonal transformation (roughly a rotation or rotoinversion) to align one point with a chosen rectangular coordinate axis. In our case, our core objects are *lists* of vectors and we desire an orthogonal transformation to ensure that all elements of that list have distinct y coordinates. In order to apply it, we just need to prove that such an orthogonal transformation exists. This is fairly straightforward because there are infinitely many possible angles of rotation in the plane and only finitely many pairs of vertices in the list.


```

|- ∀p:(real^2)list.
  ∃f q. (∃g. orthogonal_transformation g ∧ f = MAP g) ∧
        (∀x y. MEM x q ∧ MEM y q ∧ ¬(x = y) => ¬(x$2 = y$2)) ∧
        f q = p
    
```

Having achieved the desired coordinate transformation, the remainder of the informal proof can be carried through. However, even this turned out to be quite difficult. Recall that the informal proof has two cases. Once we have established the triangle ABC with B as the lowest (and *strictly* lowest, since all y coordinates are different) vertex, we either split the polygon with a line AC



or choose a point D in the interior of ABC at maximal distance from the line AC and split the polygon with the line BD



In either case, in order to apply the Jordan Triple Curve Theorem, we need to establish that this new cut has non-empty intersection with the inside of the original polygon and does not intersect the polygon itself except at the endpoints. Simply establishing the first requirement is not trivial, despite its intuitive obviousness. Essentially, to get started we need to establish some point that is in both the inside of the polygon and the interior of the triangle ABC . Once that is achieved, it is fairly straightforward in either case to argue that we can join this point to at least some point on the new line segment without touching the polygon itself.


We construct a point known to be inside the polygon and the interior of the triangle ABC as follows. First, we pick points A' on AB and C' on AC (neither of them being B) such that of the vertices in the original polygon, only B has a smaller y -coordinate than either A' or C' . After establishing the the triangle $A'BC'$ is not degenerate (that is, those three points are affinely independent), we can easily show that there is a point in its interior. Intuitively, we believe it is also in the inside of the polygon because there are no other points inside $A'BC'$ to 'interfere'. To show this formally, we use a kind of parity lemma. Roughly speaking, this asserts that given any simple closed curve including a line segment from a to b , if another segment from c to d crosses the line segment from a to b exactly once and does not otherwise intersect the curve, then c and d have opposite 'inside' and 'outside' status.

```

|- ∀ a b c d p x:real^2.
  simple_path(p ++ linepath(a,b)) ∧
  pathstart p = b ∧ pathfinish p = a ∧
  segment(a,b) INTER segment(c,d) = {x} ∧
  segment[c,d] INTER path_image p = {}
  ⇒ (c IN inside(path_image(p ++ linepath(a,b))) ⇔
     d IN outside(path_image(p ++ linepath(a,b))))
  
```

It is easy to establish that points below our vertex B are outside the polygon, and then to use this parity lemma to deduce that our constructed point lies inside. This still leaves the task of proving that our new cut does not intersect the original polygon except at its endpoints. In the first case above, this is relatively easy, but the second is a bit more involved. The idea of picking the new vertex D as far from the line AC as possible (as opposed to, for example, the lowest vertex in the interior of ABC) is to avoid cases like

the following where another vertex V manages to ‘sneak inside’ so that our new cut BD does after all intersect the polygon.



The geometric reasoning underlying the fact that our choice of vertex D indeed ensures non-intersection is in principle straightforward, but we found it quite hard work reasoning about obvious facts like ‘this point and that point are on opposite sides of the line’. Nevertheless, we do ultimately obtain our goal of proving that a polygon can always be chopped into two pieces with a cut as required. Note that the condition that the length of the vertex list is at least 5 corresponds to the fact that the polygon has at least 4 vertices (since the first and last elements of the vertex list are the same).

```
|- ∀p:(real^2)list.
    simple_path(polygonal_path p) ∧
    pathfinish(polygonal_path p) = pathstart(polygonal_path p) ∧
    5 <= LENGTH p
    => ∃a b. ¬(a = b) ∧ MEM a p ∧ MEM b p ∧
        segment(a,b) SUBSET inside(path_image(polygonal_path p))
```

As a result, we can rather easily drive an inductive proof based on the number of vertices to obtain the final Pick theorem:

```
|- (∀x. MEM x p => integral_vector x) ∧
    simple_path(polygonal_path p) ∧
    pathfinish(polygonal_path p) = pathstart(polygonal_path p)
    => measure(inside(path_image(polygonal_path p))) =
        &(CARD {x | x IN inside(path_image(polygonal_path p)) ∧ integral_vector x}) +
        &(CARD {x | x IN path_image(polygonal_path p) ∧ integral_vector x}) / &2 -
        &1
```

4. Conclusions and related work

A presentation by Narboux *et al.*[†] discusses a Coq proof of part of the theorem, but leaves a number of gaps and as far as we know has not been completed. So, as far as we are aware, the work we describe here is the first successful formalisation of Pick's Theorem. Nevertheless, it was somewhat chastening to discover how difficult it was to formalise some apparently elementary pieces of intuitive geometric reasoning. The main culprit seems to be the concept of 'inside', which is a very natural notion but difficult to reason about. It would be a very appealing project to try to automate more of it, perhaps by a more searching analysis of our everyday intuition.

Our general approach to Pick's Theorem was, naturally enough, chosen because we believed that it would be relatively easy to formalise. In fact, the proof up to the case of an arbitrary lattice triangle worked very nicely, and was streamlined by the use of the inclusion–exclusion result for additive functions. As one of the reviewers pointed out, this is reminiscent of the inclusion–exclusion expansion over faces in the Euler–Poincaré formula and some of its common proofs (Barvinok 2002; Webster 1995). We believe that the same sort of reasoning would easily generalise to the case of a convex polygon. Indeed, one proof we looked at in the literature and were considering adopting for our formalisation (Murty and Thain 2007) actually proves the theorem only for a convex lattice polygon.

All in all, the main difficulties are connected with the treatment of arbitrary polygons and their 'inside' and 'outside'. In this domain, many intuitively obvious facts require a considerable amount of work to prove rigorously, and it is often difficult to find a thorough informal treatment to use as a model without digging into fairly old literature such as Lennes (1911). It may be that we made things more difficult for ourselves by generalising where possible from polygons to arbitrary paths, but our hope is that in compensation the background lemmas are more likely to be useful elsewhere. Here again, we needed to explore older literature to find proofs of results like the 'Jordan Triple Curve theorem'. As noted, we took our proof from Whyburn (1964). The same result is discussed in Newman (1939), but it is proved by a modified form of a particular proof of the Jordan Curve Theorem, not deduced directly from the result itself. A restricted form for polygons, which would have sufficed for our present purposes, is proved in Thomassen (1992).

One naturally wonders whether a radically different proof of Pick's Theorem might avoid all the difficulties associated with triangulation of polygons, perhaps using results such as the Gauss–Green theorem relating line and plane integrals. Some intriguingly different proofs are given in Kurogi and Yasukura (2005), Diaz and Robins (1995) and Blatter (1997), all of which may generate interesting formalisation challenges. It would also be natural to consider formalising extensions and generalisations of Pick's Theorem, and other related results. One possibility is to extend it to non-simple polygons using Euler characteristics (Dubeau and Labbé 2007). Another is to consider higher-dimensional

[†] 'Vers une preuve formelle du théorème de Pick en Coq', <http://galapagos.gforge.inria.fr/December2008/mns-galapagos-dec-2008.pdf>.

generalisations using Ehrhart polynomials (Ehrhart 1967). Another interesting future direction would be to use Pick's Theorem as a case study for automated theory exploration, since the concepts it involves are combinatorial and finite, lending themselves to automated conjecture formation and testing.

Acknowledgements

The author is grateful to Tom Hales for help with the Jordan Triple Curve Theorem proof, to the referees for their corrections and insightful suggestions, and to the audiences at several talks on the subject for additional ideas.

References

- Barvinok, A. (2002) *A Course in Convexity*, Graduate Texts in Mathematics **54**, American Mathematical Society.
- Blatter, C. (1997) Another proof of Pick's area theorem. *Mathematics Magazine* **70** 200.
- Diaz, R. and Robins, S. (1995) Pick's formula via the Weierstrass \wp -function. *The American Mathematical Monthly* **102** 431–437.
- Dubeau, F. and Labbé, S. (2007) Euler's characteristics and Pick's theorem. *International Journal of Contemporary Mathematical Sciences* **2** 909–928.
- Ehrhart, E. (1967) Sur un problème de géométrie diophantienne linéaire II. *Journal für die reine und angewandte Mathematik* **227** 25–49.
- Hales, T. C. (2007a). Easy pieces in geometry. (Available at <http://www.math.pitt.edu/~thales/papers/>.)
- Hales, T. C. (2007b). The Jordan curve theorem, formally and informally. *The American Mathematical Monthly* **114** 882–894.
- Harrison, J. (1996) HOL Light: A tutorial introduction. In: Srivas, M. and Camilleri, A. (eds.) Proceedings of the First International Conference on Formal Methods in Computer-Aided Design (FMCAD'96). *Springer-Verlag Lecture Notes in Computer Science* **1166** 265–269.
- Harrison, J. (2005) A HOL theory of Euclidean space. In: Hurd, J. and Melham, T. (eds.) Theorem Proving in Higher Order Logics, 18th International Conference, TPHOLs 2005. *Springer-Verlag Lecture Notes in Computer Science* **3603** 114–129.
- Harrison, J. (2009a). A formalized proof of Dirichlet's theorem on primes in arithmetic progression. *Journal of Formalized Reasoning* **2** (1) 63–83.
- Harrison, J. (2009b). Formalizing an analytic proof of the Prime Number Theorem (dedicated to Mike Gordon on the occasion of his 60th birthday). *Journal of Automated Reasoning* **43** 243–261.
- Harrison, J. (2009c). Without loss of generality. In: Berghofer, S., Nipkow, T., Urban, C. and Wenzel, M. (eds.) Proceedings of the 22nd International Conference on Theorem Proving in Higher Order Logics, TPHOLs 2009. *Springer-Verlag Lecture Notes in Computer Science* **5674** 43–59.
- Kurogi, T. and Yasukura, O. (2005) From Homma's theorem to Pick's theorem. *Osaka Journal of Mathematics* **42** 723–735.
- Lenne, N. J. (1911) Theorems on the simple finite polygon and polyhedron. *American Journal of Mathematics* **33** 37–62.
- Murty, M. R. and Thain, N. (2007) Pick's Theorem via Minkowski's theorem. *The American Mathematical Monthly* **114** 732–736.

- Newman, M. H. A. (1939) *Elements of the Topology of Plane Sets of Points*, Cambridge University Press.
- Pick, G. (1899) Geometrisches zur Zahlenlehre. *Sitzungsberichte des Deutschen Naturwissenschaftlich-Medicinischen Vereines für Böhmen 'Lotos' in Prag, Series 2* **19** 311–319.
- Thomassen, C. (1992) The Jordan-Schoenflies theorem and the classification of surfaces. *The American Mathematical Monthly* **99** 116–130.
- Webster, R. (1995) *Convexity*, Oxford University Press.
- Whyburn, G. T. (1964) *Topological Analysis*, Princeton Mathematical Series **23**, Princeton University Press, revised edition.